Внимание! Эта статья была опубликована в журнале “КомпьютерПресс” 10’99 на прилагаемом к нему компакт-диске в файле VBSTRING.HTM. В разделах, связанных с национальными особенностями преобразования кодов символов,  использовались символы из различных  кодовых таблиц (Cyrillic, Western и Central European). К сожалению, при преобразованиях форматов файлов со статьей в окончательном варианте (VBSTRING.HTM) искажений изображения символов. Это касается разделов “Еще несколько любопытных моментов” и “Как перекодировать символы в Word 97” (включая таблицу 3).

Правильное изображение всех примеров приведено в данном файле и символы, где была допущена ошибка выделена шрифтом  “16, жирный, курсив”.

ВНИМАНИЕ! Данный файл предназначен для загрузке только в среде Word 97/2000. При его загрузке в Word 6.0 может опять произойти ошибки преобразования символов!

Особенности работы со строковыми переменными в VB

Часть 1

Андрей Колесов

Программный код многих приложений связан с обработкой стоковых переменных. Visual Basic всегда отличался достаточно большим набором функций для выполнения таких операций, однако их эффективное применение требует от разработчика хорошего понимания внутреннего механизма их реализации. На этот аспект проблемы уже не раз обращалось внимание в наших "Советах для тех, кто программирует на VB" (например, см. Советы 11 и 135) <*Сноска 1>.

<*Сноска 1. Список публикаций в рубкике “Советы для тех, кто программирует на VB” с указанием номеров советов и номеров журнала “КомпьютерПресс” можно найти по адресу www.visual.2000.ru/develop/vb/compress/index.htm”.> 
Строки переменной и фиксированной длины 

VB использует строковые переменные двух типов: с переменной длиной (до 2^31 байт) и с фиксированной длиной (до 2^16 байт):

Dim VariableLength$ ' строка переменной длины 

Dim VariableLength As String ' строка переменной длины 

Dim FixedLength As String*8 ' строка фиксированной длины (=8)

Обратите внимание, что тип строки переменной длины можно задать с помощью суффикса $. Так было принято еще в первых версиях Бейсик и, на наш взгляд, данное представление более удобно для работы. Но это, конечно, дело привычки и вкуса.

Принципиальным отличием строк двух этих типов является порядок формирования их содержимого <**Сноска2>. В момент создания строки переменной длины оператором Dim в программе резервируется только внутренний описатель переменной (в начальный момент длина строки равна нулю), а память под содержимое переменной динамически выделяется (и заполняется соответствующим значением) только в момент выполнения операции присвоения:

VariableLength$ = Строковое выражение или переменная

<**Сноска2: Здесь нужно еще помнить об отличии формирования статических и динамических переменных. Первые формируются при компиляции кода, а вторые -- в процессе выполнении программы. Но вопрос работы со статическими и динамическими данными не входит в тему настоящей статьи.>

Область памяти для содержимого строки фиксированной длины резервируется однозначно при ее первом определении и заполняется в начальный момент нулевым цифровым кодом.

VB-программисты обычно пользуются строками переменной длины, так как это позволяет не задумываться о необходимой длине, а также экономить память (выделяется столько места, сколько действительно нужно). Однако при этом следует помнить, что операции с любыми динамическими переменными по определению требуют больше времени по сравнению со статическими, поэтому при работе со строковыми переменными данный тезис особенно актуален.

Ключевым положением является то, что в отличие от работы с данными фиксированной длины (практически всеми простыми типами данных, в том числе числовыми) самая длительная операция с динамическими строками -- это самое простое присвоение: A$=..., так как в этом случае каждый раз идет обращение к внутреннему диспетчеру динамической памяти, который фактически формирует новую переменную. Например, сравнение таких двух операторов:

VariableLength = "Андрей"

FixedLength = "Андрей"

покажет, что второй выполняется примерно в 2,5 раза быстрее, чем первый. В этой связи можно сформулировать такой СОВЕТ: при работе со строками переменной длины избегайте операций присвоения. Подробный пример реализации такой рекомендации подробно рассмотрен в нашем Совете 135. Здесь же рассмотрим два варианта решения задачи со строкой длиной 255 символов, каждый байт которой равен его номеру в строке.

Вариант 1:

a$ = "": For n% = 1 To 255: a$ = a$ & Chr$(n%): Next

Вариант 2:

a$ = Space$(255): For n% = 1 To 255:_

Mid$(a$,n%) = Chr$(n%): Next

Второй вариант выглядит более длинным, но он будет работать пример в два раза быстрее. Причем если увеличивать длину формируемых байтов, то разница в скорости будет расти до десятков и даже сотен раз. Этот алгоритм можно использовать и для случая, когда длина строки до начала цикла ее формирования неизвестна -- сначала резервируется строка заведомо большей длины, а после окончания цикла выделяется нужное количество символов оператором Left$.

Передача строковых параметров в DLL

Слеудет сразу отметить, что передача строковых переменных в VB-процедуры (внутри данного проекта) и в DLL выполняется по-разному.

При работе со строками необходимо иметь в виду различия в их формировании в VB и других системах программирования (С, Pascal, Fortran и др.). В VB для каждой строковой переменной формируется ее описатель, в котором хранится адрес самой строки и ее длина. Этот описатель и передается в качестве параметра при передаче данных между процедурами. В С строка передается с помощью непосредственно ее адреса (указателя), а ее длина символов определяется по нулевому значению байта (код ASCII = 0) в конце сроки. Таким образом, в C нулевой код не может использоваться в качестве значимого символа внутри строки. <***Сноска 3>
<***Сноска 3. Язык C использует несколько иной механизм динамического выделения памяти для строковых переменных. За счет некоторой избыточности резервирования области памяти под переменную повышается скорость работы со строками при изменении их длины.>

Обычные DLL, в том числе и Win API, используют модель языка C при работе со строковыми переменными. Поэтому для передачи строки в такую DLL необходимо использовать ключевое слово ByVal. В этом случае автоматически создается новая строка, в конце которой приписывается нулевой код. В саму библиотеку передается адрес новой строки LPSTR. (Следует также помнить о преобразовании из кода Unicode в ANSI, о чем было писано в Совете 203.)

Такие функции не могут напрямую создавать строковые переменные в формате VB, однако они способны выполнять преобразование переданных им строк, значение которых возвращается в вызывающую программу. Рассмотрим для примера API-функцию, которая определяет значение системного каталога Windows. Типичное обращение к ней выглядит следующим образом:

Declare Function GetSystemDirectory Lib "kernel32" Alias _ 
"GetSystemDirectoryA" (ByVal lpBuffer As String, _ 


ByVal nSize As Long) As Long 


' 


' Резервируется буфер, достаточный для 


' размещения полного имени каталога 


BufferLength& = 256 


Buffer$ = Space$(BufferLength) 


' Для обеспечения надежной работы API-функции 


' в качестве параметра передается длина буфера 


' 


Result& = GetSystemDirectory(Buffer$, BufferLength&) 


If Result& > BufferLength& Then 



MsgBox "Ошибка! Недостаточная длина буфера!" 


Else 



DirName$ = Left$(Buffer$, Result&) 


End if

В данном случае функция сама возвращает число загруженных символов имени каталога. Но можно было бы обойтись и без значения Result&, написав такой код:

LenName& = Instr(Buffer$, Chr(0)) 

If LenName& > 0 Then 


DirName$ = Left$(Buffer$, LenName& - 1) 

End If

В приведенном примере Buffer$ используется только для возврата данных. Но возможен случай, когда переменная используется и для передачи, и для новой строки. Тогда можно предложить такой вариант:

BufferLength& = 256 

' буфер заполняется нулевым кодом 

Buffer$ = String$(BufferLength, 0) 

StartName$ = "Andrei" 

Mid$(Buffer$) = StartName$ 

Call SomeApiFuction(..., ByVal Buffer$, ...) 

Result& = Instr(Buffer$, Chr(0)) 

ResultName$ = Left$(Buffer$, Result& - 1)

Здесь резервируется строка длиной 256 символов, но с точки зрения функции SomeApiFuction ее начальное значение имеет только 6 знаков.

При обращении к VB-процедуре в качестве параметра передается исходный описатель строки. Если же используется вариант ByVal, то фактически просто создается копия переменной, которая обрабатывается внутри вызванной процедуры.

При обращении к DLL строковая переменная может передаваться и без использования ключа ByVal. Однако это возможно делать только тогда, когда данная библиотека реализована в варианте OLE 2.0. В этом случае туда передается указатель в формате VB (BSTR) и DLL-процедура работает с ним примерно так же, как и VB-процедура.

ВНИМАНИЕ. Традиционное использование ключевого слова ByVal означает одностороннюю передачу данных в вызываемую процедуру. Это происходит и при обращении к VB-процедурам (для переменных любого типа, в том числе и String), и при обращении к DLL-процедурам (для любых переменных, за исключением String). Соответственно при передаче строкой переменной в DLL в режиме ByVal ее значение в вызывающей программе может быть изменено.

Обработка строк в VB

Для обработки строк в VB имеется довольно большой набор встроенных операторов и функций. Однако среди них можно выделить несколько базовых, с помощью которых получаются все остальные. Ключевыми функциями являются: операция конкатенации (слияния) строк, операции сравнения (равно, больше, меньше), Mid$ (оператор), Mid$ (функция), Asc и Chr$.

Вот примеры реализации некоторых встроенных функций:

Function MyLeft$ (MySting$, NumLeft%) 


' выделить NumLeft левых символов 


MyString$ = Mid$(MyString$, 1, NumLeft%) 

End Function

Function MyLtRim$ (MySting$) 


' убрать левые пробелы 


Dim NumS% 


Do 



NumS% = NumS% + 1 


Loop While Mid$(MyString$, NumS%, 1) = " " 


MyLtRim$ = Mid$(MyString$, NumS%, Len(MyString$) _



- NumS% +1) 

End Function

Раньше, до версии VB 2.0, для операции конкатенации использовался только знак "+" (плюс). Однако в VB 3.0 стало можно также применять и знак "&" (амперсанд). Их принципиально отличие заключается в том, что "+" подразумевает наличие переменных и выражений только строкового типа, а "&" допускает любые типы (они автоматически преобразуются в строковые).

Например, оператор

Symbol1$ = Symbol2$ + 10

приведет к появлению ошибки (недопустимый тип операнда). А оператор

Symbol1$ = Symbol2$ & 10

выполнится без проблем. Хотя для конкатенации Microsoft рекомендует использовать второй вариант, отметим, что следующий эквивалентный код:

Symbol1$ = Symbol2$ + Str$(10)

не намного сложнее, но позволяет более точно следить за преобразованиями типов данных. С точки зрения быстродействия я не обнаружил сколь-нибудь значительных отличий в этих вариантов.

Будьте осторожно с кодом 0 

Ранее уже говорилось о том, что VB позволяет использовать любые значения кода байтов от 0 до 255. И тем не менее в ряде случаев он "плохо" реагирует на некоторые коды, в частности на 0, который он воспринимает как конец строки (чего он не должен делать). Например, напишите такой код с использование текстового поля:

Start$ = "Петя" + Chr$(0) + "Коля" 

Text1.Text = Start$ 

Result$ = Text1.Text 

Print Start$; Len(Start$) 

Print Text1.Text; Len(Text1.Text) 

If Text1.Text <> Finish$ Then Stop 'остановки не будет 

If Start$ <> Text.Text Then Stop ' здесь будет остановка

Вы получите такие результаты:

Петя Коля 9 Петя 4

Здесь видно, что текстовое поле восприняло код 0 как конец строки и обрезало ее. В переменную Finish$ было записано уже измененное значение строки. Получается, что в результате тривиальной операции присвоения происходит изменение строки.

Использование байтовых строк

Переменные типа String традиционно называются "строковыми", при этом подразумевается, что в них содержится последовательность байтов с любым числовым значением от 0 до 255. Здесь следует еще раз подчеркнуть, что VB в отличие от C, позволяет использовать байт с кодом 0.

Использование строковых переменных для хранения и передачи данных произвольной структуры бывает очень удобно. Обычно для описания структуры данных применяется тип User Defined, например, так:

Type MyType 


int1 As Integer 


long1 As Long 

End Type 

Dim MyDate As MyType

Однако подобную структуру можно легко представить в виде строки:

MyString$ = Mki$(MyDate.int1) + Mkl$(MyDate.long1)

К сожалению, Microsoft не включила операторы Mkx$ в состав VB (они же были в ее Basic для DOS!), но их достаточно легко создать самому пользователю с помощью функций API (см. Совет 173). Мне сейчас не хотелось бы отвлекаться на обсуждение того, зачем нужно и когда полезно использовать строки вместо структур -- это отдельная тема. Сошлюсь на собственный опыт. Такой технический прием обеспечил нам (когда мы еще писали "боевые" коммерческие программы) полную независимость программного кода от конкретной структуры данных: наши программы могли работать без перекомпиляции кода с любыми структурами таблиц и любыми конфигурациями визуального интерфейса.

Хотелось бы отметить, что для такого варианта (строка двоичных байтов) все операции сводились к слиянию или разделению отдельных поле строки, преобразованию их в переменных другого формата (например, Integer или та же String) и последующей обработке этих отдельных данных.

Еще в VB4 были также введены переменные типа Byte (беззнаковые целые числа 0-255). В этой связи нужно в первую очередь отметить, что скорость обработки (пересылка, арифметические и логические операции) целочисленных переменных различного типа (Byte, Integer и Long) практически одинакова в современных процессорах <****Сноска4>. Поэтому преимущество использования байтовых переменных проявляется в первую очередь при хранении информации, т.е. когда речь идет не о простых переменных, а о массивах.

<**** Сноска 4: Intel отказалась от блока оптимизации своих процессоров для 16-разрядных операций при переходе от Pentium к семейству P6. Во многом именно поэтому преимущества P6 (тогда еще Pentium Pro) были хорошо видны только для 32-разрядных приложений. Для моделей AMD K5 (о K6 мы не имеем такой информации) специальная оптимизация для 8-ми и 16-разрядных чисел существовала, а поэтомув этих режимах они имели порой заметное преимущество перед Pentium II. По нашим оценкам байтовые операции в VB6 (процессоры Pentium и Cyrix 200 МГц) выполняются даже немного медленнее, чем с 2-х и 4-х байтовыми целыми. Возможно, это происходит из-за того, что данный режим обработки на самом деле эмулируется чисто программным образом.>
В свою очередь одномерный байтовый массив является практически точным аналогом строковой переменной с точки зрения хранения данных, но при этом имеет заметные преимущества в скорости обработки данных.

Рассмотрим такой, пример (Exam3.vbp): необходимо произвести инверсию всех байтов, хранящихся в переменной Source$. Тут можно предложить два варианта решения задачи:

' Source$ -- исходная строка 

' 

' Вариант 1. Работа со строкой 


For k% = 1 To LenB(Source$) 



MidB$(Source$, k%, 1) = _



ChrB$(Not AscB(MidB$(Source$, k%, 1))) 


Next 

' 

' Вариант 2. Использование временного байтового массива 


Dim arrByte() As Byte 


arrByte = Source2$ 


For k% = 0 To LenB(Source2$) - 1 



arrByte(k%) = Not arrByte(k%) 


Next 


Source2$ = arrByte

По моим оценкам второй вариант (переменная с 60 байтами) работает в 5 раз быстрее, хотя и выглядит несколько громоздко (резервирование дополнительного массива, перезапись в него байтов, потом обратная перезапись данных в строку). Причина этого вполне понятна: копирование последовательности байтов выполняется очень быстро (одной машинной командой), но при работе с байтовым массивом не нужно использовать довольно длительные операции преобразования их строковой переменной в байтовую и обратно. (Каждая такая операция сопровождается созданием временных строковых переменных, но в данном случае это свидетельствует о слабой оптимизации таких часто встречающихся конструкций создателями VB.)

Здесь стоит также отметить, что я использовал очень простой вариант копирование из строки в массив и обратно. Однако это можно применять только для динамических байтовых массивов. Кроме того, нужно обратить внимание, что при операции:

arrByte = strSource$

всегда создается массив с нумерацией индекса от 0, даже если установлен режим Option Base 1. То есть всегда создается массив такой размерности:

ReDim arrByte(0 to LenB(strSource$) - 1 )

Отметим также, что использование строковых переменных для хранения двоичных байтов может быть гораздо эффективнее, чем применение массовов Byte. Например, массив srtArray$(N) фактически представляет собой уникальную конструкцию, позволяющую выполнять прямой доступ к записям переменной длины.

Особенности работы с символьными данными 

Однако основное применение строк приходится на работу с символьными данными -- алфавитно-цифровым представлением при обмене информацией между человеком и компьютером. Более того, многие строковые функции ориентированы именно на обработку символьных данных. Например, LTrim$ убирает слева только крайние пробелы (код ASCII = 32). В то же время ряд функций поиска и сравнения могут работать со строками в двух режимах: двоичной обработки (в этом случае, например, будет Z < z) и символьной (Z = z).

Ключевая проблема заключается в том, что, с одной стороны, некоторые двоичные коды не воспринимаются как символы (например, тот же 0) или используются некоторым зафиксированным образом (коды управления печатью, например перевод строки -- 13), с другой стороны, -- 255 символов (на самом деле меньше) просто не хватает для всего многообразия символов народов мира.

Радикальным решением данного вопроса является переход от традиционной для MS DOS/Windows системы кодирования символов ANSI/DBCS к универсальной Unicode (см. врезку "От ANSI/DBCS к Unicode"). Поскольку DBCS используется только для азиатских стран, далее будем говорить только об ANSI и Unicode.)

Один из аспектов этой проблемы для VB-программистов заключается в том, что все 32-разрядные версии VB используют для внутреннего хранения Unicode, т.е. один символ занимает два байта, а не один, как ранее в 16-разрядных системах. Следовательно, понятия "байтовая строка" и "символьная строка" перестали быть синонимами. В этой связи мы обратим внимание на следующие моменты:

1. Ранее любая двоичная байтовая строка могла быть интерпретирована как некий набор ASCII-кодов (значения 0-255 были значимыми), то теперь такая интерпретация может вызвать ошибку выполнения (незначимый код). В частности, VB (и многие другие приложения) при выводе символьной информации автоматически обозначают несуществующие значения знаком вопроса (?). Проблема заключается в том, что многие американские приложения не признают других наборов символов, кроме родного cp1252, и поэтому считают символы кириллицы также ошибочными. (Наверное, многие читатели сталкивались с получением электронных писем, состоящих из одних вопросительных знаков.)

2. Многие привычные строковые функции на самом деле теперь работают только с символьными данными: Asc, Chr, Input, InStr, Left, Right, Len, Mid. Для работы с байтами нужно использовать модификации этих функций, в которых к тем же именам в конце добавлена латинская буква B: AscB, ChrB и т.д.

3. В соответствии с принятыми в VB правилами младший байт в целочисленном коде является первым по порядку, а старший -- вторым.

Проиллюстрирую сказанное на таких примерах:

Str1$ = "Петя" 

Print Len(Str1$); LenB(Str2$) 'напечатано: 4 8 

For i = 1 to Len(Str1$) 


Print Hex$(Asc(Mid(Str1$, i, 1))), _



Hex$(AscW(Mid(Str1$, i, 2)) 


' Будет напечатана таблица 


' с ANSI и Uni-кодами (шестнадцитиричными) 


' символьной строки: 


' CF 41F 


' E2 435 


' F2 442 


' FF 44F 

Next 

' 

For i = 1 to LenB(Str1$) 

Print Hex$(AscB(MidB(Str1$, i, 1))); 


' Будет напечатана последовательность 


' значений двоичных байтов: 


' 1F 4 35 4 42 4 4F 4 

Next

'

Str2$ = Str1$ + Chr$(43) + "Лена" 

Print Str2$ ' будет напечатано: Петя+Лена

'

Str2$ = Str1$ + ChrB$(43) + "Лена" 

Print Str2$ ' напечатано: Петя???? 

Print Len(Str2$); LenB(Str2$) ' напечатано: 8 17

В последнем случае я искусственно включил между двумя символьными строками один двоичный байт. В результате произошло смещение байтов (четные и нечетные поменялись местами) и коды символов для второй части строки стали незначимыми (их при выводе заменили вопросительные знаки “?”). Кроме того, число байтов в строке стало нечетным (17), что невозможно для символьной строки.

Если вы работаете со строками, используя только символьные функции, то все сказанное здесь не имеет значения. Но если вы применяете двоичные операции (например, для повышения быстродействия -- это будет показано далее), то все эти нюансы нужно иметь в виду.

Национальные особенности работы с символьными строками

Как было сказано выше, при работе со символьными переменными в VB постоянно происходит преобразование данных из ANSI в Unicode и обратно. Рассмотрим такой пример:

strValue$ = "И" ' преобразование из ANSI в Unicode Print strValue$ ' преобразование из Unicode в ANSI

Проблема здесь заключается в неоднозначности вариантов такого преобразования. Дело в том, что русскому символу "И" соответствует в ASNI/ASCII значение 200 (C8 в шестнадцатиричной системе). Однако этому же коду соответствуют символы "È" (для западноевропейских языков, кодовая таблица 1252) и "Č" (для восточноевропейских языков, кодовая таблица 1250). В тоже время все эти три символа имеют разный Unicode (шестнадцатиричное представление): И — 0418, È — 00C8, Č — 0107.

В приложениях, использующих ANSI-коды (например, Word 6.0, среда VB или Notepad), изменение изображения этого символа достигается простой заменой шрифта самим пользователем. В приложениях, использующих Unicode (Word 97), выбор изображения символа (выбор шрифта) выполняется автоматически самой программой в соответствии с кодом символа.

Итак, вопрос формулируется следующим образом: по каким правилам происходит преобразование кодов? Ответ на него может быть интересен тем, кто создает приложения для интернациональных приложений (которые могут работать не только внутри России) и тем, кто пишет макросы для многоязычных документов в Word 97.

В VB/VBA преобразование символьных кодов в явном виде выполняется двумя способами:

1. Функциями Chr/Asc, которые преобразуют числовой ASCII-код в строковую переменную (два двоичных байта, один символ) и наоборот.

a$ = Chr(200) ' один Unicode-символ, два байта 

AsCode% = Asc(a$) ' число 200

2. Функцией StrConv, которая делает такие же преобразования в зависимости от ее второго параметра, но такая операция выполняется над всеми байтами строки:

a$ = "Петя" ' строка Unicode-кодов 

b$ = StrConv(a$, vbFromUnicode) ' строка ANSI-кодов 

c$ = StrConv(b$, vbUnicode) ' строка Unicode-кодов 

Print LenB(a$); LenB(b); LenB(c$) ' 8 4 8 —  число байт 

Print (a$ = c$) ' True, переменные равны

Так вот, для VB 5.0 и VBA 5.0 (который входит в состав MS Office 97) в обоих случаях режимом преобразования управляет параметр "ANSI кодовая таблица" (ACP), который записан в системном реестре по адресу:

HKEY_LOCAL_MACHINE\System\CurrentControlSet\Contol\Nls\CodePage\

Для VB 6.0 (и, вероятно, для VBA в составе MS Office 2000) для функций Chr/Asc используется тот же параметр ACP, а для StrConv -- параметр "региональные установки" (LocaleID), который хранится в системном реестре по адресу:

HKEY_LOCAL_MACHINE\System\CurrentControlSet\Contol\Nls\Locale\

Оба этих параметра определяются в Windows 9x при начальной установке системы, но при этом могут быть изменены (в международных версиях) прямой коррекцией реестра, а LocaleID -- через окно Control Panel | Regional Setting.

Значения этих параметров в системе можно определить в программе с помощью API-функций:

Declare Function GetSystemDefaultLCID& Lib "kernel32" ()

Declare Function GetACP& Lib "kernel32" ()

Следует также отметить, что параметр ACP в VB 6.0 влияет на режим перекодировки при неявных операциях преобразования кодов типа:

strValue$ = "Петя": Print strValue$

А параметр LocaleID управляет режимом преобразования для операций UCase/LCase (и их аналога StrConv), а также при выполнении операций сравнения -- InStr, StrComp -- в режиме Text Compare.

В VB 6.0 реализован расширенный вариант функции StrConv -- можно использовать LocaleID, установленный в системе, или задать его в явном виде:

b$ = StrConv(a$, vbFromUnicode) ' системный LocaleID 

с$ = StrConv(a$, vbFromUnicode, &H419) ' Russia

Из сказанного выше можно определиться параметры приложения или операционной системы, которые влияют представление (для Win16) или преобразование (Win32) национальных символов:

Группа языков
Win16, ANSI
 Win32, Unicode




Тип шрифта
Кодовая таблица
LocaleID (региональные установки)

Central European
Courier CE
cp1250
&h415 (Польша)

Cyrillic
Courier Cyr
cp1251
&h419 (Россия)

Western
Courier
cp1252
&h409 (США)

При этом следует подчеркнуть следующие моменты:

1. При работе с приложениями, использующими ANSI формат символьных данных, тип шрифта влияет только на внешнее представление символа. Тип шрифта устанавливается индивидуально для приложения (например, Word 6.0 или среда VB 6.0).

2.  В приложениях, применяющих Unicode, кодовая таблица и региональные установки (которые являются параметрами Windows, а не приложения!) влияют на преобразование из ANSI в Unicode и обратно. 

3. Каждой кодовой таблице (и типу шрифта) соответствуют несколько национальных языков (стран). Польша, Россия и США выбраны здесь только для примера.

Еще несколько любопытных моментов 

Все обстоит достаточно просто при преобразовании из ANSI в Unicode. Выполнение следующего кода приведет к формированию разных двоичных строк в зависимости от установленной кодовой таблицы:

strValue$ = "И" 

Print Hex$(AscW(strValue$))

Для cp1250 будет напечатано 10C, для cp1251 -- 418, для cp1252 -- C8. В то же время нужно иметь в виду специфику выполнения такого кода:

strValue1$ = Chr(200) 

strValue2$ = StrConv(ChrB(200), vbUnicode) 

Print (strValue1$ = strValue2$)

При работе в среде VB 5.0 переменные strValue1$ и strValue2$ будут всегда равны, в среде VB 6.0 -- только в случае согласованных значений кодовой таблицы и региональных установок.

Гораздо более запутанно обстоит дело с преобразованием из Unicode в ANSI. Выполним такой программный код:

v1250$ = ChrW(&H010C) 

v1251$ = ChrW(&H0418) 

v1252$ = ChrW(&H00C8)

Здесь формироваются три переменную, состоящие из одного символа  — соответственно Č, И, È, —  каждой из которых соответствует один и тот же ANSI-код -- 200 (&h00C8). Однако, что же получится, если выполнить далее такую операцию:

Print Asc(v1250$); Asc(v1251$); Asc(v1252$)

Казалось бы, должно получиться "200 200 200", но на самом деле это не так — результат будет выглядеть следующим образом:

Переменная 

(Unicode, изображение)


Результат выполнения функции Asc для различных кодовых таблиц, 

десятичное/шестнадцатиричное представление


cp1250
cp1251
cp1252

v1250$ (&h010C, Č)
200/C8
63/3F
69/45

v1251$ (&h0418, И)
67/43
200/С8
69/45

v1252$ (&h00C8, È)
67/43
63/3F
200/C8

Из этого можно сделать такой вывод: Правильное преобразование из Unicode в ANSI выполняется только в том случае, когда код таблицы Unicode (старший байт-код) соответствует кодовой таблице (или параметру региональных установок для функции StrConv в VB6). Возможно, Microsoft так и хотел сделать, но такая задумка выглядит весьма странно — вполне вероятно в данном случае допущена просто ошибка.

Может показаться, что вся приведенная здесь информация имеет скорее напоминает теоретические размышления, однако практическое применение всего этого проиллюстрировано во врезке "Как перекодировать символы в Word 97".

Следует также упомянуть о том, что в составе Win API имеется целый набор фунций работы со строковыми переменными, но в целом он не очень значительно расширяет возможности встроенных VB-функций. Пользоваться последними гораздо проще.

Выводы и советы

1. Использование оптимальных конструкций при работе строковых данных позволяет порой увеличить скорость обработки в десятки раз. При этом внешне простые констукции порой работают медленнее, чем более сложные и вроде бы, громоздские.

2. Если вам критична производительность — не поленитесь проветсти специальные тестовые испытания разных вариантов программных конструкций.

3. Использование двоичной обработки строковых переменных чаще всего позволяет повысить скорость операций, но при этом  нужно предельно внимательным.

4. При обработки символьных данных нужно учитывать национальные особенности алгоритмов и влияние тех или иных системных установок. Помните, что это относится не только к внутренним переменным, но и к идентификаторам, передаваемым, например, через OLE Automatiom. Поэтому рекомендую использовать для имен объектов, их свойств и методов, только английские символы.

VBP-проекты и Word-документов примеров , приведенных в данной статье, вы можете найти на Web-узле по адресу: http://visual.2000.ru/develop/vb/source/.

(Окончание следует. В следующей части будет рассказано о новых строковых функциях VB 6.0, операторе Like, и использовании байтовых переменных для ускорения операций работы со строками.)

ДОПОЛНИТЕЛЬНЫЕ ВРЕЗКИ (2штуки)

ОТДЕЛЬНАЯ ВРЕЗКА 1

От ASCII к ANSI/DBCS и Unicode

Проблема оптимального кодирования символов, наверное, берет свое историческое начало с создания телеграфа, где применялось 4-х битовое обозначение. 5-й разряд на перфоленте был контрольным. Для перехода от одного набора символов к другому использовались специальные символы "смены регистра". На смену этому коду пришел 7-битовый (8-й -- контрольный), который позволил одному языку (английскому или русскому) умещаться полностью. Поэтому именно он положил основу системе ASCII (American Standard Code for Information Interchange) еще задолго до создания современных компьютеров.

Переход на "байтовый" принцип хранения информация произошел только в конце 60-х годов, но тем не менее в компьютерной технике еще долго доминировал 7-разрядный код (американских разработчиков тогда не очень волновала проблема завоевания зарубежных рынков, и уже тем более -- адаптации к национальным языкам).

В результате "локализация" ПО для зарубежных стран (для СССР, в частности) решалась простой заменой строчных английских букв прописными русскими. Таким образом, до конца 80-х годов советские программисты работали только с верхним регистром английского и русского алфавита.

Решением проблемы использования национальных языков стал переход к системе ANSI (American National Standards Institute) и DBCS (Double-Byte Character Set). ANSI является сегодня наиболее популярным символьным стандартом для ПК, в котором используется 8-битовое кодирование (до 256 кодов). Основу его составляет традиционная (основная) таблица ASCII (0-127), дополненная еще 128 символами, которые иногда называют "расширенной" таблицей ASCII. Именно поэтому ANSI часто называют расширенным вариантом ASCII (или даже в качестве синонима ACSII, что на самом деле не совсем верно).

ANSI применяется в основном для европейских языков, которые основаны на классическом латинском алфавите, но в отличие от английского языка имеют некоторые дополнительные символы, а еще чаще -- варианты расширений типа "A с точкой", "I с двумя точками" и пр.

Русский язык (кириллица) также смогла уместиться в дополнительный набор символов (33 буквы), но с некоторыми проблемами. Дело в том, что многие коды из состава таблицы 128-255 уже использовались (без соответствующей стандартизации) в целом ряде программ для некоторых специальных целей (американским разработчикам не приходило в голову тогда, что их программы будут использоваться в России). В качестве классических примеров можно вспомнить русскую "р", которую не признавал Norton Commander 3.0 для DOS, или FoxBase, который считал русскую "Н" переводом строки.

Таких "тяжелых" наследий прошлого было довольно много и именно этим во много объясняется наличие специальных "панъевропейских версий" Windows до 95 включительно -- адаптация к использованию подобных нестандартизованных спецсимволов требовала специальной коррекции кода.

Однако многим языкам (в основном азиатским) дополнительных 128 символов никак не хватало. Поэтому появилась система DBCS, которая строится по другому принципу. Для латинских букв она использует кодировку ASCII (0-127). Для своих собственных символов (японский набор включает около 12 тыс. символов) она использует двухбайтовый набор, когда первый код (128-255) выполняет функцию упреждающего, а следующий является его дополнением. Легко посчитать, что такая кодировка позволяет использовать 32K+128 символов.

UniCode -- это схема кодирования символов, которая использует на каждый символ два байта: первый байт определяет как бы номер набора символов, второй -- собственно код символа. Например, набор символов английского и западноевропейских языков (cp1252) имеет код первого байта 0, восточноевропейских (cp1250) -- 1, русского (кириллица, cp1251) -- 4.

Эта система продвигается Международной организацией по стандартизации (ISO) в качестве универсального единого стандарта. И если говорить о программном обеспечении компьютеров, то она сейчас переживает непростой период перехода от ANSI/DBCS к Unicode, когда в силу огромного информационного наследства приходится использовать смешанные варианты кодирования. Именно этим во многом определяется значительная путаница с преобразованием кодов символов, о которой мы говорим сейчас на примере VB.

Типичным примером является Microsoft Windows 9x, ядро и WinAPI которой по-прежнему используют кодировку ANSI/DBCS. Однако многие ее компоненты, а также автономные приложения (в частности, VB и Office) применяют Unicode (см. таблицу 1).

Подчеркну, что хотя Unicode и DBCS имеют двухбайтовые символы и покрывают набор символов примерно одного объема, их схемы кодирования абсолютно различны. С точки зрения программистов, принципиальным моментом является следующее: ANSI использует строго один байт на символ, Unicode -- строго два байта, а DBCS -- 1-2 байта, в зависимости от используемых символом.

Кроме того, известной проблемой русского языка является наличие несколько однобайтовых кодировок символов (см. таблицу 2). Строго говоря, ANSI кодировкой для кириллицы является таблица Windows cp1251. 

Таблица 1. Использование схем кодирования символов в MS Windwos
Компоненты Windows
Используемые наборы символов 

Visual Basic
Unicode 

32-разрядные библиотеки объектов
Unicode 

16-разрядные библиотеки объектов
ANSI/DBCS 

Windows NT API
Unicode 

Windows 9x API
ANSI/DBCS 

Automation в Windows NT
Unicode 

Automation в Windows 9x API
Unicode

Таблица 2. Шестнадцатиричные коды русских букв в различных кодовых

таблицах для кириллики (Cyrillic)

Символ
Кодовые таблицы 


Win(cp1251)
Unicode
KOI8-R
DOS,cp866

(альтернативная)
IS08859-5

(основная)
Mac

А
C0
0410
E1
80
B0
80

Б
C1
0411
E2
81
B1
81

В
C2
0412
F7
82
B2
82

Г
C3
0413
E7
83
B3
83

Д
C4
0414
E4
84
B4
84

Е
C5
0415
E5
85
B5
85

Ж
C6
0416
F6
86
B6
86

З
C7
0417
FA
87
B7
87

И
C8
0418
E9
88
B8
88

Й
C9
0419
EA
89
B9
89

К
CA
041A
EB
8A
BA
8A

Л
CB
041B
EC
8B
BB
8B

М
CC
041C
ED
8C
BC
8C

Н
CD
041D
EE
8D
BD
8D

О
CE
041E
EF
8E
BE
8E

П
CF
041F
F0
8F
BF
8F

Р
D0
0420
F2
90
C0
90

С
D1
0421
F3
91
C1
91

Т
D2
0422
F4
92
C2
92

У
D3
0423
F5
93
C3
93

Ф
D4
0424
E6
94
C4
94

Х
D5
0425
E8
95
C5
95

Ц
D6
0426
E3
96
C6
96

Ч
D7
0427
FE
97
C7
97

Ш
D8
0428
FB
98
C8
98

Щ
D9
0429
FD
99
C9
99

Ъ
DA
042A
FF
9A
CA
9A

Ы
DB
042B
F9
9B
CB
9B

Ь
DC
042C
F8
9C
CC
9C

Э
DD
042D
FC
9D
CD
9D

Ю
DE
042E
E0
9E
CE
9E

Я
DF
042F
F1
9F
CF
9F

а
E0
0430
C1
A0
D0
E0

б
E1
0431
C2
A1
D1
E1

в
E2
0432
D7
A2
D2
E2

г
E3
0433
C7
A3
D3
E3

д
E4
0434
C4
A4
D4
E4

е
E5
0435
C5
A5
D5
E5

ж
E6
0436
D6
A6
D6
E6

з
E7
0437
DA
A7
D7
E7

и
E8
0438
C9
A8
D8
E8

й
E9
0439
CA
A9
D9
E9

к
EA
043A
CB
AA
DA
EA

л
EB
043B
CC
AB
DB
EB

м
EC
043C
CD
AC
DC
EC

н
ED
043D
CE
AD
DD
ED

о
EE
043E
CF
AE
DE
EE

п
EF
043F
D0
AF
DF
EF

р
F0
0440
D2
E0
E0
F0

с
F1
0441
D3
E1
E1
F1

т
F2
0442
D4
E2
E2
F2

у
F3
0443
D5
E3
E3
F3

ф
F4
0444
C6
E4
E4
F4

х
F5
0445
C8
E5
E5
F5

ц
F6
0446
C3
E6
E6
F6

ч
F7
0447
DE
E7
E7
F7

ш
F8
0448
DB
E8
E8
F8

щ
F9
0449
DD
E9
E9
F9

ъ
FA
044A
DF
EA
EA
FA

ы
FB
044B
D9
EB
EB
FB

ь
FC
044C
D8
EC
EC
FC

э
FD
044D
DC
ED
ED
FD

ю
FE
044E
C0
EE
EE
FE

я
FF
044F
D1
EF
EF
DF









Ё
A8
0401
B3
F0
A1
DD

ё
B8
0451
A3
F1
F1
DE

Как перекодировать символы в Word 97 

Преобразование кодов в Windows выполняется довольно странно

Наверное, многие из читателей встречались с ситуацией, когда в результате копирования или перезаписи текстов неожиданно происходили такие странные преобразования текста:

из

Привет читателям!

в

Ïðèâåò ÷èòàòåëÿì!
а потом в

I?eaao ?eoaoaeyi!
а еще в какой-то момент в

????????????????!
Проиллюстрирую причины таких неожиданных преобразований на примере MS Office и VBA. Однако следует отметить, что обнаруженные здесь проблемы характерны не только для VB и их причина заключается в специфике работы системных функций Windows API.

Для выполнения данного примера вам понадобится Word 6.0 и Word 97.

Загрузите обе программы. В документе Word 6.0 введите три одинаковые строки с подобным набором символов:

АБВГДЕЖЗИЙК

Вы видите нормальные русские символы, если у вас установлен какой-то шрифт кириллицы, например, Courier Cyr (Courier -- чтобы были символы одной ширины). Теперь выделите вторую строку текста установите для нее шрифт Courier (западноевропейский набор или Western), а для третей строки -- Courier CE (восточноевропейский, который по-английски называется "центральноевропейский"). Вы увидите такой текст:

АБВГДЕЖЗИЙК 

ÀÁÂÃÄÅÆÇÈÉÊ
ŔÁÂĂÄĹĆÇČÉĘ

Сохраните документ как DocWord6.doc. Теперь сохраните этот же документ в формате "Просто текст" (DocWord6.txt) и закройте. А теперь заново прочитайте DocWord6.txt и вы увидите:

АБВГДЕЖЗИЙК

АБВГДЕЖЗИЙК

АБВГДЕЖЗИЙК

Ситуация понятна: в этом случае мы имеем дело с тремя строками, в которых каждый символ задан однобайтовыми ANSI-кодами со значениями 128-138. Двоичные коды байтов во всех строках одинаковы и их графическое изображение зависит от установки пользователем типа шрифта. В TXT-файле мы просто убрали форматирование -- установили одинаковый шрифт для всех строк. Обратите внимание, что формирование содержимого TXT-файла никак не зависит от установленной кодовой таблицы или региональных установок. (TXT-файлы лучше посмотреть простым редактором типа Notepad).

Теперь скопируйте через буфер обмена содержимое DocWord6.doc в новый документ DocWord8.doc и далее проделайте описанные выше манипуляции по созданию и чтению TXT-файла. В DocWord8.doc вы увидите те же строки:

АБВГДЕЖЗИЙК 

ÀÁÂÃÄÅÆÇÈÉÊ 
ŔÁÂĂÄĹĆÇČÉĘ

а вот в DocWord8.txt будет нечто новое:

АБВГДЕЖЗИЙК 

AAAAAA?CEEE 

RAAAALCCCEE

В данном случае мы имеем дело с такой ситуацией: в результате последовательности преобразования ANSI -> Unicode -> ANSI для двух последних строк мы получили, что результирующий вариант не соответствует исходному. Чего, в общем-то, не должно быть.

Такое преобразование данных объясняется следующим образом. При копировании строк из Word 6.0 в Word 97 символьные данные были преобразованы из ANSI в Unicode, при этом учитывался установленный тип шрифта -- Script -- для каждой строки (то есть соответствующая ему кодовая таблица -- 1251, 1252 и 1250). Таким образом три одинаковые двоичные строки байтов ANSI-кодов были преобразованы в три РАЗНЫЕ строки Unicode-кодов, а те в свою очередь -- в РАЗНЫЕ строки ANSI-кодов.

Обратное же преобразование из Unicode в ANSI (при сохранении в виде TXT-файла) было произведено по следующим правилам:

1. Для русского кода исходный ANSI-код был восстановлен правильно.

2. Для символов других языков коды верхней ASCII-таблицы (128-255) были преобразованы в 7-разрядное представление по правилу, когда расширенные варианты латинских символов, которые используются в европейских языках, заменяются на их основной вариант (например, разные варианты A -- с крышкой, точкой, апострофом и пр. -- заменяются на английскую A).

Правильное преобразование для русского кода объясняется тем, что в данном случае у нас в Windows 98 была установлена кодовая таблица кириллицы, cp1251. В случае cp1252 мы получили DocWord8.txt (Script=Cyrillic):

??????????? 

АБВГДЕЖЗИЙК 

RБВAДLCЗCЙE

а для cp1250 (Script=Cyrillic):

???????????

AБВAДAAЗEЙE

АБВГДЕЖЗИЙК
Результаты проведенных преобразований приведены в таблице 3. Обратите внимание, что при чтении TXT-файла в Word 97 происходит обратная перекодировка из ANSI в Unicode в соответствии с текущей системной кодовой таблицы, то есть автоматически выбирается нужный тип шрифта Script. Поэтому для сравнения получаемых результатов в TXT-файле лучше пользовать редактором NotePad и устанавливать одинаковый тип шрифта.

Здесь мы видим, что символы, у которых нет "базового" аналога из таблицы кодов 32-127, производится их замена на "?" (&h3F). Причем это касается не только символов кодовой таблицы кириллицы, но и европейского символа Æ (Unicode -- &h00C6).

Конечно, ситуация, когда правильное преобразование из Unicode в ANSI выполняется только для кодов, которые соответствуют кодовой таблице, установленной в Windows, представляется если не ошибочной, то, по крайней мере очень странной. Ведь в этом случае происходит потеря информации (разные исходные ANSI-коды заменяются одинаковыми): обратное восстановление исходного кода невозможно.

ПРИМЕЧАНИЕ. В ходе подготовки этой статьи была обнаружена  ошибочная ситуация в работе Word 97. Когда я прочитал в Word 97 созданный ранее файл DocWord6.doc, то увидел такое его содержимое:

АБВГДЕЖЗИЙК 

АБВГДЕЖЗИЙК 

ŔÁÂĂÄĹĆÇČÉĘ

То есть вторая строка из западноевропеской записи превратилась в русскую. Это к вопросу о том, что от простого чтения  в разных версиях Word содержимое документов может меняться...
Таблица 3. Последовательность преобразования символьных данных с использованием разных кодовых таблиц.

Тип кодов
Коды символов (шестнадцатиричная система)


А
Б
В
Г
Д
Е
Ж
З
И
Й
К

Word 6.0 (однобайтовая ANSI-кодировка символов )

АБВГДЕЖЗИЙК

(Cyrillic) =

ÀÁÂÃÄÅÆÇÈÉÊ
 (Western)  =

ŔÁÂĂÄĹĆÇČÉĘ
(Central European)


C0
C1
C2
C3
C4
C5
C6
C7
C8
C9
CA

Word 97 (двухбайтовая Unicode-кодировка символов )

АБВГДЕЖЗИЙК 
0410
0411
0412
0413
0414
0415
0416
0417
0418
0419
041A

ÀÁÂÃÄÅÆÇÈÉÊ
00CO
00C1
00C2
00C3
00C4
00C5
00C6
00C7
00C8
00C9
00C1

ŔÁÂĂÄĹĆÇČÉĘ
0154
00С1
00С2
0102
00С4
0139
0106
00С0
010С
00С9
0118

Преобразование из Word 97 в TXT для разных кодовых таблиц

(однойбайтовая ANSI-кодировка символов)

cp1251 (Cyrillic)

АБВГДЕЖЗИЙК
C0
C1
C2
C3
C4
C5
C6
C7
C8
C9
CA

AAAAAA?CEEE
(английские символы)
41
41
41
41
41
41
3F
43
45
45
45

RAAAALCCCEE
(английские символы)
52
41
41
41
41
4С
43
43
43
45
45

cp1252 (Western)

???????????
3F
3F
3F
3F
3F
3F
3F
3F
3F
3F
3F

ÀÁÂÃÄÅÆÇÈÉÊ
(Western)   =

АБВГДЕЖЗИЙК
(Cyrillic)
C0
C1
C2
C3
C4
C5
C6
C7
C8
C9
CA

RÁÂAÄLCÇCÉE
 (Western)   =

RБВAДLCЗCЙE

(Cyrillic)
52
C1
C2
41
C4
4C
43
C7
43
C9
45

cp1250 (Сenrtal European)

???????????
3F
3F
3F
3F
3F
3F
3F
3F
3F
3F
3F

AÁÂAÄAAÇEÉE
(CE)        =

AБВAДAAЗEЙE

(Cyrillic)
41
C1
C2
41
C4
41
41
C7
45
C9
45

ŔÁÂĂÄĹĆÇČÉĘ

(CE)        =

АБВГДЕЖЗИЙК
(Cyrillic)
C0
C1
C2
C3
C4
C5
C6
C7
C8
C9
CA

Как восстановить текст в Word 97 

К сожалению, случай, когда довольно неожиданно происходит преобразование типа:

АБВГДЕЖЗИЙК -> ÀÁÂÃÄÅÆÇÈÉÊ
встречается довольно часто. В Word 6.0 восстановление исходного текста производится очень просто -- установкой нужного шрифта. Но в Word 97 такой возможности нет: выбор шрифта (Cyrillic или Western) делается автоматически в зависимости от Unicode-кода. Как быть?

Для этого можно воспользоваться VBA и написать соответствующую макрокоманду, но на самом деле встроенных средств VBA недостаточно для полного решения такой задачи. Проблема заключается в том, что он может правильно преобразовывать из Unicode в ANSI только символы установленной в системе в данный момент кодовой таблицы, то есть по тем же правилам, что описаны выше.

Решение задачи на VB 6.0 выглядело бы следующим образом:

a$ = Selection.Text

RegionTo% = &H419  ‘ в какой код преобразовывать

For i% = 1 To Len(a$)


k1% = AscW(Mid(a$, i, 1))  ‘ Unicode


k2% = k1%\&h100            ‘старший байт кода


Select Case k2%   ‘ в зависимости от типа кодовой таблицы 

    Case 4  ‘ Cyrillic




RegionFrom% = &H419   ‘Россия



Case 1  ‘Central European




RegionFrom% = &H415   ‘Польша



Case 0  ‘ Western 




RegionFrom% = &H409   ‘США



Case Esle ‘Все остальные 




Stop   ‘неизвестный код


End Select

  a1$ = StrConv(ChrW(k1%), vbFromUnicode, RegionFrom%)

Mid$(a$, i, 1) = StrConv(a1$, vbUnicode, RegionTo%)

Next i

Selection.Text = a$

Но для Word 97 оно не годится -- в VBA 5.0 нет расширенного варианта функции StrConv. Хотя, возможно, в Office 2000 эта функция уже реализована. Тем не менее вариант преобразования из Western в Cyrillic (наиболее распространенный вариант искажения текстов) решается и в Word 97 достаточно просто:

a$ = Selection.Text

For i% = 1 To Len(a$)


k1% = AscW(Mid$(a$, i, 1))


If k1% > 255 Then ‘ код Cyrillic 



a1$ = ChrW$(k1%) 


Else ‘ код Western (ANSI=Unicode)



a1$ = Chr$(k1%)


End If


Mid$(a$, i, 1) = a1$

Next i

Selection.Text = a$

Тут я воспользовался тем, что для символов Western коды Unicode и ANSI равны. Впрочем, и для всех других случаев проблема перекодировки является вполне разрешимой -- нужно просто самим составить таблицу Unicode-кодов для разных кодовых таблиц и написать достаточно простую программу их преобразования.

